
News  
Spike2 and Signal Training Days 

Our US Spike2 and Signal training courses will take place on Thursday 1st and Friday 2nd 
of November, prior to the Society for Neuroscience meeting in San Diego. The venue will 
be the Holiday Inn San Diego – On The Bay. These training days are for both existing 
and prospective users of Spike2 and Signal and are suitable for beginners and advanced 
users. There is also our popular Workshop thread where you can discuss your particular 
requirements with a CED engineer or programmer in a small group environment. 

If you or any of your colleagues would like to attend, please follow the links to view the 
training day program and complete the registration form. Please give a short description 
of the type of research that you do and analysis that you are interested in. This will allow 
us to tailor sessions for specific requirements. 

NEW! Power1401 mk II 

 
We are pleased to announce the release of the Power1401 mk II, our latest high-
performance laboratory interface. Below is a comparison with the Power1401 625: 

Power1401 mk II Power 1401 625kHz 
1MHz multi-channel sampling rate 625kHz multi-channel sampling rate 

  Intel 800MHz XScale® processor  233MHz RISC processor  

  256MB – 1GB memory 32MB – 256MB memory 

 

Another feature of the Power1401 mk II, due to customer feedback over the years, is that 
the on/off switch is now a push button located on the front of the unit. This means that 
users no longer have to crawl behind their hardware rigs to turn the 1401 on or off! 

Meetings and events 
European Brain and Behaviour Society (EBBS)                                                                                                                   
Miramare Campus 
Trieste, Italy  
September 15th – 19th 2007 

18th Ion Channel Meeting                                                                                                           
Presqu’île de Giens 
France  
September 23rd – 26th 2007 

Polish Neuroscience Society - 8th International Congress                                                                                                       
Jagiellonian University 
Krakow, Poland 
September 24th – 27th 2007 

      NEW! Hints 

 
When you see this icon, 
double-click it to reveal extra 
hints or tips relating to the 
current newsletter article. 

http://www.ced.co.uk/nw9y.htm
http://www.ced.co.uk/tdapply.htm
http://www.ebbs-science.org/
http://congres.igh.cnrs.fr/canaux-ioniques/index2007.htm
http://www.krakowconvention.com/ptbun


Italian Neuroscience Society 2007                                                                                   
Palazzo della Gran Guardia 
Verona, Italy 
September 27th – 30th 2007 

International Society of Autonomic Neuroscience 2007                                                                
Kyoto Terrsa 
Kyoto, Japan 
October 5th – 8th 2007 

 
Latest versions of Spike2 and Signal 

Updates for Spike2 and Signal are available from the CED downloads page, or by 
clicking on the links in the table below. Demonstration versions of the latest software are 
also available.   

 Spike2 downloads Signal downloads 
Spike2 version 6.05  Signal version 3.10 

 Spike2 version 5.18  Signal version 2.16 

 Spike2 demo  Signal demo 

Spike2   

 

Q. I am recording EMG data using Spike2 and using a separate telemetry system to 
capture strain gauge data. The telemetry system does not give any analogue 
output and is not synchronized to the Spike2 sampling. This leads to a time drift 
between the two files. Is there any way to synchronize the timing of them for 
analysis? 

A. There is a script command, FileTimeBase(), that can be used to change the 
 underlying base time of a Spike2 data file, stretching or shrinking the time axis for 
 all data items. The main use of this function is to compensate for  time drift  
 between data sampled by Spike2 and data captured by some other non-
 synchronized external equipment. The essential requirements are that the data 
 format from the external equipment can be imported into Spike2 and the data 
 recorded from both systems should have a common marker at the start and end 
 of the recording. Spike2 can import a wide range of file formats via the Import 
 command in the File menu. Once the file is in the Spike2 format, it can be 
 modified to match an existing files timebase. 

 It is likely that there will be a time offset between the start markers in each file. 
 This can be eliminated by using the Export as… function from the File menu to 
 export the data  to a new data file and setting the checkbox to time shift the data 
 so that the first  exported range starts at 0. The  first exported range should be 
 the time of the start marker in the file, which could be found by positioning a 
 cursor on the marker, for example.  

 
Export dialog  

Back to top

Hint:  

http://www.sins.it/Index.asp
http://isan2007.jtbcom.co.jp/
http://www.ced.co.uk/uplsu.shtml
http://www.ced.co.uk/s2wu6u.shtml
http://www.ced.co.uk/sigw3u.shtml
http://www.ced.co.uk/s2wu5u.shtml
http://www.ced.co.uk/sigw2u.shtml
http://www.ced.co.uk/s2demu.shtml
http://www.ced.co.uk/sigdemu.shtml
CED
To create a file with more than 32 channels, click the Channels… button at the bottom of the Export dialog.



Next we can work out the time scaling between the two files by getting the 
interval between the start and end markers. This is the result of the time of the 
end marker minus the time of the start of the file, which could be found using 
cursor positions and the Cursor Regions dialog. If this interval is t1 for the file 
recorded by Spike2 and t2 for the imported file, we can then modify our imported 
file to have the same timebase as the Spike2 recorded file by making the 
imported file the current view and using the following code in the Script menu 
Evaluate… window. Simply replace t1 and t2 with the interval values measured 
from the data files: 

FileTimeBase(FileTimeBase()*t1/t2); 

The two files could then be amalgamated into one using the ChanSave() 
command (see eNews 9 or the on-line help for further details). 

 

  

 

Q. The analysis software that I am presently using does not support the detection of 
more than one action potential in a single sampling sweep. A colleague has 
recommended Signal as being able to do this and I would like some further 
information. 

A. Signal can import a wide range of data file formats, meaning that you can make 
use of features in Signal that are not available in other third party software. There 
is a script available which will detect and take measurements from multiple 
spikes in a single sweep (see eNews 26 for details).  

 We hope to have this functionality implemented as a built-in feature in Signal in 
the very near future using Active Cursors, which will give even greater flexibility 
and allow a number of different measurements to be taken from multiple action 
potentials both on-line and off-line. 

 

 Scripts: Spike2  

Q. I would like an automated method of generating cross correlations between 
templated spikes, where each spike template is used as a trigger for all other 
spike templates in turn. Could this be achieved with a script? 

A. The attached script, CrossChanCorrelation.s2s, can be used to generate 
multiple spike correlation histograms for all combinations of templated spikes in a 
channel. Currently it allows for up to 8 spike templates per channel (including 
code 0) for correlating between channels in the file. 

 The script is controlled from a toolbar where the user can open a data file for 
analysis and control the parameters for the correlation as well as the number of 
correlation result views to display at any one time. 

 

 Scripts: Signal  

Q. I have been using the script for spike measurements that was included in 
eNewsletter 26 and have found it extremely useful. I would now like to be able to 
generate a phase-plane plot based on the time derivative of the membrane 
voltage versus the membrane voltage. Is this possible? 

A. The attached script, Phase-plane plot.sgs, generates a phase-plane plot in 
an XY view based on all action potentials in a user-defined time range which can 
be set using cursors. The plot updates dynamically so the user can click and 
drag the cursor positions to update the display. 

Back to top

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

Hint:  

CED
Supported file types for import include Axon .DAT and .ABF files, .TXT files and many more. See the Import data… command in the Signal file menu for a complete list.


'$CrossChanCorrelation.s2s|Script to generate multiple spike correlation histograms for all combinations of spike types in a channel
'Currently allows for up to 10 spike types for one channel autocorrelation or 8 spikes per channel (including code 0) for correlating between channels 

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 17/09/2007

var data%;								'Data file to analyse
var chan%;								'Wavemark channel
var trigChan%;							'Trigger channel
var codes%[2];							'Number of spike types
var sp%[17];	
var spikes%[50][2];					'Array of spike codes	
var sChans%[50][2];					'Array of channels containing spikes
var cor%[1024];                     'Array of correlation views
var bins%;                            'No of bins in correlation
var numHists%;                      'No of result views
var bSize := 0.0025;                'Default binsize
var histWidth := 0.5;               'Default width for correlation
var offset := 0.25;                 'Default offset
var doRaster%;                      'Flag for raster display
var doZeroFlag% := 0;              'Flag to include 00 templates
var doAutoCorr% := 0;             'Flag for autocorrelation 

var dispWin%;                       'Current display window
var c%;
var start% := 0;                    

var setWinL[64];                'Co-ordinates for display of multiple result views
var setWinT[64];
var setWinR[64];
var setWinB[64];


View(App(3)).WindowVisible(0);      'Hide the script

'Main toolbar
ToolbarSet(1, "Quit");
ToolbarSet(2, "Open file", Open%);
ToolbarSet(3, "Correlate", DoCorr%);
ToolbarSet(4, "Close results", CloseResults%);
ToolbarSet(6, ">>",ShowNext%);
ToolbarSet(7, "<<",ShowPrev%);
ToolbarSet(9, "Display", SetDisp%);
ToolbarEnable(3,0);
ToolbarEnable(4,0);
ToolbarEnable(6,0);
ToolbarEnable(7,0);
Toolbar("",1023);


Func Open%()
if data% > 0 then       'If data file already open
	View(data%);
	FileClose(0,-1);       'Close it
	data% := 0;
endif;
data% := FileOpen("",0,0);      'Prompt user to select a file
if data% > 0 then
	Window(0,0,100,100);        'Maximise
	FrontView(data%);             
	ToolbarEnable(3,1);     'Enable correlation function
endif;
start% := 0;
return 1;
end;


Func DoCorr%();
var ok%;
var delChans%;
var i%,j%;
var num%;
var histFlag%;
var wTitle$;
var crossFlag% := 0;
var scale%;
var col%;

View(data%);        'Makes sure data is the current view
DlgCreate("Correlation parameters");        'Correlation settings
DlgChan(1, "Channel to analyse",18);
DlgChan(2, "Trigger channel",18);
DlgReal(3, "Histogram width",0.001,100);
DlgReal(4, "Bin size",0.0001,1);
DlgReal(5, "Offset", 0,100);
DlgCheck(6, "Include code 00");
DlgCheck(7, "Autocorrelate spikes (Single channel only)");
DlgCheck(8, "Show rasters");
DlgList(9, "Y scaling", "Count|Spikes/s",2);
DlgInteger(10, "Number of windows to display",1,64);
ok% := DlgShow(chan%,trigChan%,histWidth,bSize,offset,doZeroFlag%,doAutoCorr%,doRaster%,scale%,dispWin%);
if ok% then
	if chan% = trigChan% then       'Check for cross correlation
		crossFlag% := 0
	else
		crossFlag% := 1;
	endif;
	if doRaster% then               'Check for raster data enable
		histFlag% := 2;
	else
		histFlag% := 0;
	endif;
	bins% := histWidth/bSize;       'Calculate no. of bins required
    
	if chanKind(chan%) = 6 then     'If channel to analyse is a wavemark 
		codes%[0] := GetSpikeTypes(chan%,sp%[],0);      'Get all spike types
		ArrConst(spikes%[][0],sp%[]);
	else
		codes%[0] := 1;
		spikes%[0][0] := 0;
		sChans%[0][0] := chan%;     'Otherwise set all to 0
	endif;
	if ChanKind(trigChan%) = 6 then    'If trigger channel is a wavemark
		if crossFlag% then
			codes%[1] := GetSpikeTypes(trigChan%,sp%[],1);  'Get spike types
			ArrConst(spikes%[][1],sp%[]);
		else
			ArrConst(sChans%[][1],sChans%[][0]);    
			ArrConst(spikes%[][1],spikes%[][0]);
			codes%[1] := codes%[0];
		endif;
	else
		codes%[1] := 1;
		spikes%[0][1] := 0;
		sChans%[0][1] := trigChan%;     
	endif;
	if not crossFlag% then                      'Check for autocorrelation settings
		for i% := 0 to codes%[0]-1 do
			if not doAutoCorr% then
				for j% := i%+1 to codes%[1]-1 do
                    if spikes%[i%][0] <= 8 then
                        col% := Colour(spikes%[i%][0] + 18);        'Setup histogram colours to match spike templates
                    else
                        col% := Colour(spikes%[i%][0] + 10);
                    endif;
                    cor%[num%] := SetEvtCrl(sChans%[i%][0],bins%,bSize,offset,sChans%[j%][1],histFlag% + scale%);   'Create correlation view
					YAxisMode(3);
					FontSet(2,6,0);
					XScroller(0);
                    ChanColour(1,1,col%);
                    ChanColour(1,2,col%);
					Process(0,View(-1).MaxTime(),1,1);
					wTitle$ := Print$("%d   x   %d",spikes%[i%][0],spikes%[j%][1]); 'Set title to show spikes used
					WindowTitle$(wTitle$);
					num% += 1;
				next;
			else
				for j% := i% to codes%[0]-1 do
                    if spikes%[i%][0] <= 8 then
                        col% := Colour(spikes%[i%][0] + 18);
                    else
                        col% := Colour(spikes%[i%][0] + 10);
                    endif;
					if j% = i% then
						cor%[num%] := SetEvtCrl(sChans%[i%][0],bins%,bSize,offset,sChans%[j%][0], histFlag% + scale%);      'Set correlations
					else
						cor%[num%] := SetEvtCrl(sChans%[i%][0],bins%,bSize,offset,sChans%[j%][1], histFlag% + scale%);
					endif;
					YAxisMode(3);
					FontSet(2,6,0);
					XScroller(0);
                    ChanColour(1,1,col%);
                    ChanColour(1,2,col%);
					Process(0,View(-1).MaxTime(),1,1);
					wTitle$ := Print$("%d   x   %d",spikes%[i%][0],spikes%[j%][1]);
					WindowTitle$(wTitle$);
					num% += 1;
				next;			
			endif;
		next;
	else
		for i% := 0 to codes%[0]-1 do
			for j% := 0 to codes%[1]-1 do
                if spikes%[i%][0] <= 8 then
                    col% := Colour(spikes%[i%][0] + 18);
                else
                    col% := Colour(spikes%[i%][0] + 10);
                endif;
				cor%[num%] := SetEvtCrl(sChans%[i%][0],bins%,bSize,offset,sChans%[j%][1],histFlag% + scale%);       'Set correlations
				YAxisMode(3);
				FontSet(2,6,0);
				XScroller(0);
                ChanColour(1,1,col%);
                ChanColour(1,2,col%);
				Process(0,View(-1).MaxTime(),1,1);
				wTitle$ := Print$("C%d S%d   x   C%d S%d",chan%,spikes%[i%][0],trigChan%,spikes%[j%][1]);
				WindowTitle$(wTitle$);
				num% += 1;
			next;
		next;									
	endif;			
	numHists% := num%;
	Tile();
	if numHistS% > dispWin% then
		ToolbarEnable(6,1);
	endif;
	delChans% := Query("Delete duplicated channels?");
	if delChans% then
		View(data%);
		for j% := 0 to 1 do
			for i% := 1 to codes%[j%] do 
				if ChanKind(sChans%[i%-1][j%]) = 6 then
					ChanDelete(sChans%[i%-1][j%]);
				endif;
			next;
		next;
	endif;
	ToolbarEnable(4,1);
endif;
return 1;
end;

Func GetSpikeTypes(sChan%,types%[],c%)	;				'Function to count number of spike types in channel and put codes into array
var startCode%;
var i%;
var num% := 0;												'Number of types found 
var spikeMark%;											'Current type
var count%;													'Number of spikes found 
View(data%);
ToolbarText("Checking for all possible spike types");
MarkMask(sChan%, 0);                  		'set mode 0
MarkMask(sChan%,-1, 1,-1);            		'include everything (reset)
if doZeroFlag% then startCode% := 0; else startCode% := 1;endif; 
for spikeMark% := startCode% to 255 do					'Loop through each possible code			
	MarkMask(sChan%, 0, 0, -1);           	'exclude everything in layer 0
	MarkMask(sChan%, 0, 1, spikeMark%); 		'include the codes we want
'	DrawAll();
	count% := Count(sChan%,0,MaxTime());		'Count number of spike on channel
	if count% > 0 then									'If any found		
		types%[num%] := spikeMark%;					'Put code into array	
		sChans%[num%][c%] := ChanDuplicate(sChan%);
		ChanShow(sChans%[num%][c%]);
		num% += 1;			
		if num% > 17 then
			Message("Cannot analyse more than 17 spikes\nScript closing");
			for i% := 1 to num% do
				ChanDelete(sChans%[i%-1][c%]);
				MarkMask(sChan%,-1, 1,-1);            		'include everything (reset)
			next;
			Halt;
		endif;
	endif;
next;
MarkMask(sChan%,-1, 1,-1);            		'include everything (reset)
ToolbarText("");
return num%;
end;


Func CloseResults%()
var i%;
for i% := 1 to numHists% do     'Cycle through open result views
	View(cor%[i%-1]);
	FileClose(0,-1);            'and close them
next;
View(data%).WindowVisible(1);   'Bring data back to the front
ToolbarEnable(4,0);
return 1;
end;


Func SetDisp%()
var ok%;
var i%;
DlgCreate("Display parameters");
DlgInteger(1, "Number of windows to display",1,64);
ok% := DlgShow(dispWin%);       
if ok% then
	for i% := 1 to numHists% do     
		View(cor%[i%-1]).WindowVisible(0);
	next;
	start% := 0;
	Tile();
	CheckButtons();
endif;
return 1;
end;


Func ShowNext%()    'Display the next n correlation windows
start% += c%;       
Tile();
CheckButtons();
return 1;
end;

Func ShowPrev%()    'Display the previous n correlation windows
start% -= c%;
Tile();
CheckButtons();
return 1;
end;


Proc CheckButtons()     'Enable or disable toolbar buttons dependent on current display
if start% > 0 then
	ToolbarEnable(7,1);
else
	ToolbarEnable(7,0);     
endif;
if start% + dispWin% >= numHists% then
	ToolbarEnable(6,0);
else
	ToolbarEnable(6,1);
endif;
end;

Proc Tile();            'Setup window displays
View(data%).WindowVisible(0);
var i%;
var dh%[64];
if numHists% > 0 then       'Cycle through windows
	TileWindows(dispWin%);            'tile depending on numbers and co-ordinates
	for i% := 0 to dispWin% - 1 do
		dh%[i%] := cor%[i% + start%];
		View(dh%[i%]);
		Window(setWinL[i%], setWinT[i%], setWinR[i%], setWinB[i%]);
		FrontView(dh%[i%]);
	next;
endif;
end;


'Calculate coordinates of n tiled windows...
'Only goes up to 64 (setMaxToShow%)...
proc TileWindows(n%)
var r%;
docase
	case n% <  4 then r% := 1; c% := n%;
	case n% =  4 then r% := 2; c% := 2;
	case n% =  5 then r% := 2; c% := 3;
	case n% =  6 then r% := 2; c% := 3;
	case n% =  7 then r% := 2; c% := 4;
	case n% =  8 then r% := 2; c% := 4;
	case n% =  9 then r% := 3; c% := 3;
	case n% = 10 then r% := 3; c% := 4;
	case n% = 11 then r% := 3; c% := 4;
	case n% = 12 then r% := 3; c% := 4;
	case n% >= 13 and n% <= 15 then r% := 3; c% := 5;
	case n% = 16 then r% := 4; c% := 4;
   case n% >= 17 and n% <= 20 then r% := 4; c% := 5;
	case n% >= 20 and n% <= 24 then r% := 4; c% := 6;
	case n% = 25 then r% := 5;c% := 5;
	case n% >= 26 and n% <= 30 then r% := 5; c% := 6;
	case n% >= 31 and n% <= 36 then r% := 6; c% := 6;
	case n% >= 37 and n% <= 42 then r% := 6; c% := 7;
	case n% >= 43 and n% <= 48 then r% := 6; c% := 8;
	case n% = 49 then r% := 7; c% := 7;
	case n% >= 50 and n% <= 56 then r% := 7; c% := 8;
	case n% >= 57 and n% <= 64 then r% := 8; c% := 8;
	else r% := 0; c% := 0;
endcase;
SetWindows(r%, c%);
end;


proc SetWindows(r%, c%)
var origX;
var origY;
var row%, col%;
var index%;

index% := 0;
for row% := 0 to r%-1 do
	for col% := 0 to c%-1 do
		origX := col% * (100.0 / c%);
		origY := row% * (100.0 / r%);
		index% := index% + 1;
		setWinL[index% - 1] := origX;
		setWinT[index% - 1] := origY;
		setWinR[index% - 1] := origX + (100.0 / c%);
		setWinB[index% - 1] := origY + (100.0 / r%);
	next;
next;
end;



CED
CrossChanCorrelation.s2s


'$Phase-plane plot.sgs|Script to generate a phase plane plot of action potentials

'The script is a 'work in progress' and is offered without guarantees.
'You must test it to see whether it is suitable for your application.

'CED 17/09/2007

var file%;							'Data file handle
var plot%;							'Duplicate window handle
var sTime,eTime;					'Start and end of display section
var presTime,preeTime;				'Current start and end time of duplicate display	
var frame%,frameDup%;				'Frame numbers of data and duplicate windows
var maxPoints%;
var numPoints%;
var sItem%;
var sPoint;
var ch% := 1;							'Data channel to plot
var bSize;


ToolbarSet(0, "Cursor view",MoveView%);		'Link to idle routine to track changes
ToolbarSet(1, "Quit");						'Toolbar buttons
ToolbarSet(3, "Open File",Open%);
Toolbar("Window to show area between cursors",1023);


Func Open%()						'Function to open a data file
var ok%;
file%:=FileOpen("",0,0);			'Open data file	
Window(0,0,50,100);					'Position window in left half of screen
CursorSet(2);						'Set 2 cursors
CursorLabel(4,1,"Start");			'Label cursors
CursorLabel(4,2,"End");
FrontView(file%);
DlgCreate("AP channel selection");
DlgChan(1, "Channel to plot",1);
ok% := DlgShow(ch%);
if ok% > 0 then
    bSize := BinSize(ch%) * 1000.0;
	frame% := Frame();
	maxPoints%:=ChanPoints(1);
	plot%:=FileNew(12,0);
	View(plot%).Window(50,0,100,100);	'Position duplicate in right half of screen
	FrontView(plot%);
	XYSetChan(1,-maxPoints%,0,1);
	XYDrawMode(1,2,0);
	XYDrawMode(1,5,1);
    ChanTitle$(0,"dV/dt");
    ChanUnits$(1,"(Vs-1)");
    XTitle$("V");
    XUnits$("(mV)");
    WindowTitle$("Phase-plane plot");
endif;
Return 1;
End;

Func MoveView%()					'Idle routine 
if plot%>0 then						'Do nothing unless duplicate window is open 
	View(file%);					'Ensure data window is current view
	if Frame() <> frame% then
		PlotPoints();
		View(file%);
		frame% := Frame();
	endif;
	sTime:=Cursor(1);				'Get times of cursors
	eTime:=Cursor(2);
	if eTime < sTime then				'Ensure cursors are in corrent order
		CursorRenumber();
		sTime:=Cursor(1);
		eTime:=Cursor(2);
		CursorLabel(4,1,"Start");			'Label cursors
		CursorLabel(4,2,"End");
	endif;
	frame%:=Frame();				'Get frame number of data window
	if sTime <> presTime then				'If start time has changed
		if eTime = sTime then
			eTime:= sTime+bSize;
		endif;
		sPoint := sTime;
		PlotPoints();
		presTime:=sTime;	
	endif;
	if eTime <> preeTime then				'Same with end times
		if eTime = sTime then
			eTime:= sTime+bSize;
		endif;
		PlotPoints();
		preeTime:=eTime;
	endif;
endif;
Return 1;
End;


Proc PlotPoints()
View(file%);
numPoints%:=ChanRange(1,sPoint,eTime,sItem%);
var diff[numPoints%];
if numPoints% > 2 then
	ArrConst(diff[],View(file%,ch%).[sItem%:numPoints%]);
	ArrDiff(diff[]);
    ArrDiv(diff[],bSize);
	View(plot%);
	XYSetChan(1,-numPoints%);
	XYAddData(1,View(file%,ch%).[sItem%+1:numPoints%-1],diff[1:numPoints%-1]);
	Optimise(1);
endif;
end;






CED
Phase-plane plot.sgs



   

Phase-plane plot of action potentials 

Did you know…? 
The new Power1401 mk II interface now has ±5v or ±10V input and output ranges that 
can be selected using the Try1401 diagnostic software which is installed along with the 
Spike2 and Signal applications. By contrast, the Micro1401 mk II, as with previous 
Power1401 units, is set-up for ±5V or ±10V operation at time of order. 

Recent questions  
Q. We would like to use a total of three monitors to visualize our data collection 

during an experiment. The first screen would be assigned to the incoming data, 
the second to online averaging and the third to subject feedback. Is it possible to 
position the application across the three screens and position the various 
windows accordingly?   

A. Both Spike2 and Signal support multiple monitor setups. The Window()script 
command allows you to position the main application window in relation to 
multiple monitors and other window types relative to monitors or the position of 
the main application window. You can find further details and some example 
code in the on-line help page of the Window() command. Most PC’s have up to 
two available monitor outputs so you may find that a system with multiple 
graphics cards is required to use more than two monitors simultaneously. 

User group      

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

